
Introduction 
Two main modifications were made to the existing South African TIMES Model (SATIM) to be able to 

handle Monte Carlo Simulations: 

- Monte Carlo Framework development 
- Reduction of Model size 

Monte Carlo Framework 
A Monte Carlo framework was developed around the SATIM model to handle multiple simulations. 

The framework was developed in the GAMS language, the same language that the TIMES model is 

developed in. Monte Carlo Simulations involves running the model several hundred times each time 

with a different coherent set of assumptions, drawn from distributions established for each of the 

uncertain parameters. The distributions were either established via expert elicitation or from 

literature. At this stage, these samples were generated either in R or in excel or obtained directly 

from providers of sample data (e.g. population, oil price, etc.). The samples are all stored in an excel 

workbook. The samples are currently fixed in that results for simulation x will always contain the 

same data and so yield the same result. 

The process that is repeated for each simulation can be summarised as follows: 

1. Read in sample data for simulation x 
2. Prepare input files for SATIM 
3. Run SATIM 
4. Extract relevant results from SATIM, process and store 

The actual GAMS code that was written to do this is in the appendix. 

  



Reduced Model Size 
The South African TIMES model (SATIM) is a full sector optimisation model which represents the 

South African energy sectors and subsectors – both on demand and supply side. To represent a full 

year where energy demand changes during different periods of the year, 20 slices of time 

(timeslices) are used to categorise the year and the energy profiles of each sector. However, with 

higher levels of resolution (more timeslices) the computing demand increases exponentially and 

where numerous model runs are needed for analysis the computation time is large draw back. This is 

a comparison of a version of SATIM where 8 timeslices are used instead of 20 in order to improve on 

computing time.  

Figure 1 shows the demand profile for electricity in MW for the shortened timeslice profile of SATIM. 

This is compared with the regular 20 timeslices of SATIM as shown in Figure 2. 

The SATIM model is run for a reference scenario (REF) for the model with 20 timeslices and the same 

scenario in the 8 time slice version (REF 8TS).  

 

Figure 1 : The short timeslice demand profile in MW (for electricity) version of SATIM.  

 

Figure 2: The 20 timeslice demand profile in MW (for electricity) in SATIM.  

 

The power sector 
The power sector new build is shown in Figure 3.  

Figure 3: The new build capacities for the power sector for the regular profile SATIM and the version with shorter 
timeslices.  

There are some minor differences between the two models. These differences are highlighted in 

Table 1 below.  

Table 1: The difference in total capacities (in GW) between the two versions of the SATIM model. 

  2010 2015 2020 2025 2030 2035 2040 2045 2050 

Existing Coal                   

New Coal     0.0 -0.2 -0.5 -0.3 -0.6 -0.1 0.6 

OCGT diesel                   

Inland Gas Plants               0.1 -0.1 

Gas from N Moz                   

Gas Domestic Shale                   

Coastal Gas               0.8 2.0 

Nuclear                   

Solar Thermal               -1.2 -3.1 

Central Solar PV       0.0 1.3 0.6 2.0 3.7 4.6 

Wind                   

Hydro domestic                   

Pumped Storage                   

Biomass       0.4 0.5 0.5 0.5 0.1   

Imported Electricity (Hydro)                   

total 0 0 0 0.2 1.2 0.8 1.9 3.3 3.9 

% of total capacity 0% 0% -0.04% 0.29% 1.64% 0.91% 2.03% 3.19% 3.30% 

Table 2 shows the emissions from the power sector between the two models and highlights the very 

minute changes between the two versions of the model in terms of total emissions. 

Table 2: The total power sector emissions in Mt CO2 eq. This includes the contributions of NOx and CH4 

  2010 2015 2020 2025 2030 2035 2040 2045 2050 

REF 242.5 254.3 305.9 330.5 384.8 428.3 472.3 508.1 548.7 

REF 8TS 243.4 256.2 308.0 328.9 381.0 426.2 468.2 507.9 553.7 

% difference 0.4% 0.7% 0.7% -0.5% -1.0% -0.5% -0.9% 0.0% 0.9% 

 

The total model emissions for all sectors on both supply and demand are shown in Figure 4 and in 

Table 3 below.  

 

Figure 4: The total emissions by sector for South Africa between the two models.   

 

Table 3: The total emissions difference between the two models (in Mt CO2eq) 

  2010 2015 2020 2025 2030 2035 2040 2045 2050 

Power Sector 0.94 1.87 2.11 -1.58 -3.71 -2.12 -4.14 -0.15 5.05 

Refineries (incl.process)         0.03 0.05 0.05 0.01 -0.11 

Agriculture                   

Commerce 0.09 -0.23 -0.34 0.01 0.06 0.11 0.19 0.19 0.42 

Industry   -0.16 -0.22 -0.03 -0.04 -0.04 -0.04 -0.14 -0.02 

Residential 0.09 -0.06 0.09 0.24 0.24 0.26 0.35 0.43 1.14 

Transport           -0.01 -0.01 0.03 0.03 

Coal mining                   

Gas extraction                   

Gas transmission pipes                   

Total REF 418 443 515 612 740 821 892 930 1 003 

Total REF 8TS 419 444 516 611 737 819 888 931 1 010 

 

Demand side 
Overall the total final energy (all fuels) consumption is 0.05% less in the shorter timeslice model by 

2030 and this is difference increases to 0.43% by 2050.  

Tables 3 to 5 show the difference in fuel consumption patterns for the sectors which showed any 

change in fuel consumption. These changes analysed were only for those greater than 1% change. 

Industry 

Apart from 2010 industry has no major changes in fuel consumption. This may be partly attributed 

to the fact that industry has a much flatter demand profile than the other sectors and would thus be 

less impacted by a change in the time resolution of the model.  

Table 4: The total difference in % of fuels for the industrial sector. 

  2010 2030 2050 

Coal       

Natural Gas       

Diesel       

Gasoline       

Kerosens/Paraffin       

LPG 38%     

HFO -9% 3% 1% 

Biomass&Waste 7%     

Electricity       

 

 

 

 

Commercial sector 

The largest changes in consumption for the commercial sector is in natural gas and gasoline (petrol). 

The change in gasoline largely comes from ‘other’ end uses in the commercial sector – these are 

mainly office equipment and large devices/technologies not included in other end uses or in other 

sectors – such as incinerators or large scale dryers etc. Gasoline makes up approximetly 1% of the 

commercial sector. The drop in kerosene is almost for the exact same reason as for gasoline.  

Natural gas change however comes largely from the embedded trigeneration/cogeneration within 

the sector – about 670MW of cogeneration in the shorter timeslice model as opposed to about 

290MW in the regular timeslice length model.   

Table 5: The total difference in % of fuels used in the commercial sector.  

  2010 2030 2050 

Coal 4%     



Natural Gas -6% 12% 102% 

Diesel 4% -23% 20% 

Gasoline 5% -6% -63% 

Kerosens/Paraffin -11% 5% -34% 

LPG -45% -2% 10% 

HFO   -4% -2% 

Electricity       

 

Residential sector  

 

Table 6: The total % difference of fuels used in the residential sector.  

  2010 2030 2050 

Coal 9%   19% 

Natural Gas   -14% -14% 

Diesel       

Gasoline       

Kerosens/Paraffin   27% 227% 

LPG   2%   

HFO       

Biomass&Waste     1% 

Electricity     3% 

 

Discussion 
The change in the year timeslice profile largely affects electricity demand profiles of each sector, 

with some sectors having a larger change than others.  

In terms of the differences in the demand side sectors, the main changes occur in the commercial 

sector and the residential sector. The commercial sector has a representation of embedded 

generation in the form of cogeneration – gas engines producing electricity and exhaust heat being 

utilised to meet heating demands. The industrial sector has a largely flat profile, as opposed to 

residential which has a sharp peak, and has a very minor energy consumption changes between the 

model versions.  

 Overall, the demand sectors show minor changes in energy consumption patterns (as well 

as overall energy consumption), 

 the power sector has very minor changes in build investment plan to 2050, with less coal 

capacity, and more solar and gas power coming into the mix in the model version with less 

timeslices,  

 Emissions differences between the models are very small – on the order of about 1%, and 

 The time to compute the model with shorter timeslices and for a model period of 5 year 

increments to 2050 was approximately 20seconds as compared to the 2minutes and 

30seconds for the same model with 20 timeslices.  



In summary, the shorter timeslice model version of SATIM with just 8 timeslices to represent the 

year behaves very similarly to the same model run with 20 timeslices, but computes in about a 

seventh of the time.  

  



Appendix: Monte Carlo GAMS code 
 

*s=MCSimulation gdx=MCSimulation 

SETS 

* Overall sets 

  X                      simulations      /1*1000/ 

 

* SATIM sets 

  REG                    TIMES regions    /REGION1/ 

  ALLYEAR                All Years 

  T(ALLYEAR)             Active time periods in TIMES run 

  T1(ALLYEAR)                  TIMES first year 

  T2(ALLYEAR)                  TIMES second year 

  TO(ALLYEAR)                  TIMES other years 

  S                      TIMES timeslices 

  PRC                    TIMES Processes 

  COM                    TIMES Commodities 

  DEM                    TIMES Demand Commodities 

  DEM1(COM)              TIMES Demand Commodities for REGION1 

* Fuel processes 

  FUELP(PRC)             TIMES Fuel Processes 

  FUELPC(PRC)            TIMES coal processes 

  FUELPCPWR(PRC)         TIMES coal for power 

  FUELPCPWR_CB(PRC)      TIMES coal for power central basin 

  FUELPCPWR_A(PRC)       TIMES coal for power waterberg 

  FUELPG(PRC)            TIMES gas processes 

  FUELPO(PRC)            TIMES oil liquid fuel processes 

 

* Power Plant processes 

  TECHS                  TIMES power plant subsets 

  TECHC(PRC)             TIMES power plants whose cost are MC simulated 

  ETC(PRC)               TIMES Coal Plants 

  ETCA(PRC)               TIMES Coal Plants waterberg 

  ETG(PRC)               TIMES Gas Plants 

  ETGIH(PRC)             TIMES Gas Plants Shale 

  ETN(PRC)               TIMES Nuclear Plants 

  ERH(PRC)               TIMES Hydro Plants 

  ERSOLP(PRC)            TIMES PV techs Central 

  ERSOLPC(PRC)            TIMES PV techs Central 

  ERSOLPR(PRC)            TIMES PV techs Rooftop 

  ERSOLT(PRC)            TIMES CSP techs 

  ERW(PRC)               TIMES Wind Plants 

  ERB(PRC)               TIMES Biomass Plants 

* e-SAGE(cge) sets 

  FH              cge household groups 

  INS             cge household incomes 

  FS              cge sectors (for TIMES) 

  XC              cge scenarios 

  TC(ALLYEAR)     cge years 

 

* Emissions sets 

  CO2SET(COM)           Sectoral emissions 

 

* MC sim specific sets 

  FS_L(FS)        Linked subsectors ie excl com agr ele 

  FS_U(FS)        Other subsectors that are not linked ie com agr ele 

 

  COEF            Oil Price coefficient / a, b/ 

; 

ALIAS (ALLYEAR,AY), (ALLYEAR,TP); 

 

PARAMETERS 

*Overall Parameters 

  XRATE(X,T)               Real Exchange Rate R per $ 

 

* Parameters imported from a base CGE run 

  SFORE(FS,XC,TC) 

 

* Parameters imported from a base TIMES run 

  COM_PROJ(REG,ALLYEAR,COM) Useful Energy Demand from TIMES 

  COM_PROJ_BY(COM)          Useful Energy Demand in the base year 

  FUELPO_BY(PRC)            Base year fuel prices for oil products 

  TVACTELC(T,S)             Electricity output of transmission 

  TVCOMBAL(T,COM)           Annual Marginals 

  TVCOMBALELC(T,S)          Electricity marginals 

 

* Monte Carlo Simulation parameters drawn in from Excel 

* Main Drivers 

  SIM_GDP_Y(X,AY)           GDP growth 

  SIM_COM_S(X,T)           Share of commerce in GDP 

  SIM_POP(X,T)             Population 

 

* Data from Demand Model (spreadsheet-based at this stage) 

  SIM_DEMX(COM,AY)          Demand extracted from excel 

 

* Fuel Price Combined Data 

  SIM_FUELP(PRC,X,T)       Combined Fuel Price data 

 

* Technology Cost Combined Data 

  SIM_TECHC(PRC,X,T)       Combined Tech overnight Cost data 

 

* Coal Mining Costs 

  COAL_CV(PRC)             Coal calorific value 

* Coal to Power Plants Central Basin 

  SIM_CLE1(X,T)            Coal from Existing Mines in Central Basin Type 1 (conveyor) 

  SIM_CLE2(X,T)            Coal from Existing Mines in Central Basin Type 2 (rail-truck) 

  SIM_CLN(X,T)             Coal from New Mines in Central Basin 

 



* Coal to Power Plants in Waterberg 

  SIM_CLE_A(X,T)           Coal from Existing Mines in Waterberg 

  SIM_CLN_A(X,T)           Coal from new Mines in Waterberg 

 

* Coal to Synthetic Fuels 

  SIM_CLS(X,T)             Coal from Existing Mines in Central Basin to synthetic fuels 

  SIM_CLS_A(X,T)           Coal from New Mines in the Waterberg to synthetic fuels 

 

* Metallurgical Coal existing 

  SIM_COA(X,T)             Coal from Existing Mines supplying other industries 

 

* Global Commodity Prices 

  SIM_GCOAL(X,T)           Global Price for Coal in 2010 $ per ton 

  SIM_GGAS(X,T)            Global Price for Gas in 2010 $ per Mbtu 

  SIM_GOIL(X,T)            Global Price for Oil in 2010 $ per barrel 

 

* Oil products 

  OCRFAC(FUELPO,COEF)      Coefficients for mapping from oil to product prices 

 

* Natural Gas 

  SIM_GIH(X,T)             Shale Gas Production Cost $ per Mbtu 

 

* Solar PV 

  SIM_PV_Module(X,T)       PV Module Prices in 2010 $ per W 

  SIM_PV_BOS(X,T)          PV Balance of System Cost in 2010 $ per W 

  PV_CRATIO(PRC)        Ratio of BOS cost for other PV systems to Central Fixed system 

 

* Solar CSP 

  SIM_CSP(X,T)             CSP Cost in 2010 $ per W 

  CSP_CRATIO(PRC)       Ratio of reference CSP cost to other CSP techs 

 

* Nuclear Costs 

  SIM_NUCLEAR(X,T)         Overnight Nuclear Cost $ per kW 

 

* Intermediate parameters 

 

  SFORE_S(FS,AY)        Value added shares of GDP for TIMES sectors obtained from cge run 

  GDP_CGE(AY)           GDP from cge run 

  GDP_CGE_FS_S(FS,X,AY) Sector GDP shares from cge run 

  GDP_CGE_FS_S_Sum(X,AY)   Sum of linked sectors shares from cge run 

 

  GDP(X,AY)                 Overall GDP 

  GDP_FS(FS,X,AY)           Sector GDP 

  GDP_FSX(FS,AY)            Sector GDP for simulation X 

  GDP_FS_S(FS,X,AY)         Sector GDP shares 

  GDP_FS_S_Sum(X,AY)        Sum of linked sectors shares 

 

  POPX(T)                  Population for simulation X 

 

  SIM_ERSOLP(X,ERSOLP,T)   PV costs going to SATIM 

  SIM_ERSOLT(X,ERSOLT,T)   CSP costs going to SATIM 

 

* Parameters extracted from TIMES run results 

  TVCOMNET(T,COM)          Emissions from SATIM 

  TVCAP(T,PRC)             Capacity from SATIM 

  TVACT(T,PRC)             Activity from SATIM 

  TVACTELC(T,S)            Electricity output from transmission 

  TVBOMBALELC(T,S)         Electricity marginals downstream of transmission 

 

* Results kept for all simulations 

  SIMO_CO2S(CO2SET,X,T)            CO2 emissions 

  SIMO_PWRCAP(TECHS,X,T)     Capacity of power plants GW 

  SIMO_PWRACT(TECHS,X,T)     Production of power plants TWh 

  SIMO_COAACT(FUELPC,X,T)  Production of coal mines tons 

  SIMO_COATRA(X,T)         Coal Transfers from Waterberg to CB tons 

  SIMO_COAP(X,T)           Average Coal Price for power plants R per ton 

  SIMO_COAPA(X,T)          Average Coal Price for power plants in waterberg R per ton 

  SIMO_COAPB(X,T)          Average Coal Price for power plants in central basin R per ton 

 

  SIMO_GASACT(FUELPG,X,T)   Production of natural gas 

 

  SIMO_OBJz(X)             SATIM Objective function 

 

  SIMO_DMD(COM,X)          Final year demand by end-use 

 

  SIMO_PEAKELC(T,X)        Electricity peak (GW downstream of transmission) 

  SIMO_ELCDMD(T,X)         Electricity demand (TWh downstream of transmission) 

  SIMO_ELCP(T,X)           Electricity price 

* Other Parameters 

  TGAP(AY)                   TIMES Gaps 

 

 

* Set Allocation 

  PWRSET(PRC)              Power Plants set allocation 

 

; 

scalar POBjz; 

 

* File declarations 

FILE SIM_COMCOST_FILE /".\satm\Gams_WrkTI\sim_comprice+REGION1.dds"/; 

FILE SIM_PRCCOST_FILE /".\satm\Gams_WrkTI\sim_prccost+REGION1.dds"/; 

FILE SIM_DEM_FILE /".\satm\Gams_WrkTI\sim_dem+REGION1.dds"/; 

 

* Import sets and parameters from Base TIMES run 

$gdxin  .\SATM\Gams_WrkTI\Gamssave\run02_MC.gdx 

$load PRC COM DEM ALLYEAR T S COM_PROJ 

 

DEM1(COM) = DEM('REGION1',COM); 

 

* Base year demand 

COM_PROJ_BY(COM) = COM_PROJ('REGION1','2006',COM); 

 



* Import sets and parametes from Base CGE run 

* $gdxin  output1_bu.gdx 

$gdxin  simulation1.gdx 

$load XC TC FS FH INS SFORE 

 

* Set Exchange Rate (Constant for now-based on differential inflation rate differentials) 

XRATE(X,T) = 8.23; 

 

* Import MC parameters and data from spreadsheet 

$call    "gdxxrw i=MCSim.xlsx o=mcsim index=index!a6 checkdate" 

$gdxin   mcsim.gdx 

$loaddc FUELP FUELPO FUELPC FUELPCPWR FUELPCPWR_CB FUELPCPWR_A FUELPG 

$loaddc TECHS TECHC ETC ETCA ETG ETGIH ETN ERH ERSOLP ERSOLPC ERSOLPR ERSOLT ERW ERB CO2SET 

$load PWRSET 

$load SIM_POP SIM_GDP_Y SIM_COM_S 

$load SIM_GCOAL SIM_GGAS SIM_GOIL OCRFAC 

$load COAL_CV SIM_CLE1 SIM_CLE2 SIM_CLN SIM_CLE_A SIM_CLN_A 

$load SIM_GIH 

$load PV_CRATIO SIM_PV_MODULE SIM_PV_BOS CSP_CRATIO SIM_CSP SIM_NUCLEAR 

 

* Preparation of cge input data 

  FS_L(FS) = YES; 

  FS_L('agr') = no; 

  FS_L('elec') = no; 

  FS_L('com') = no; 

 

  FS_U(FS) = not FS_L(FS); 

 

 

*Count gap between TIMES years 

set TA(ALLYEAR)            TIMES years /2006*2050/; 

T1(T)$(ORD(T) EQ 1) = YES; 

T2(T)$(ORD(T) EQ 2) = YES; 

TO(T) = yes; 

TO(T1) = no; 

 

TGAP(AY)$(T(AY) AND NOT T1(AY)) = ORD(AY) - SMAX(TP$T(TP), ORD(TP)$(ORD(AY) GT ORD(TP))); 

 

* set inital and following years' shares equal to shares in cge run 

GDP_CGE(TC) = sum(FS, SFORE(FS,'base',TC)); 

SFORE_S(FS,TC) = SFORE(FS,'base',TC)/GDP_CGE(TC); 

*GDP_FS_S(FS,X,TO) = 1; 

GDP_CGE_FS_S(FS,X,TO) = SFORE_S(FS,TO); 

GDP_CGE_FS_S(FS,X,T1) = GDP_CGE_FS_S(FS,X,'2010'); 

 

* Calculate GDP shares for all simulations 

GDP(X,'2006') = 1; 

 

*loop(T$(NOT T1(T)), 

loop(TA$(NOT T1(TA)), 

*  GDP(X,T) = GDP(X,T-1)*((1+SIM_GDP_Y(X,T))**TGAP(T)); 

  GDP(X,TA) = GDP(X,TA-1)*((1+SIM_GDP_Y(X,TA))); 

 

); 

 

 

* set all shares equal to that of cge 

GDP_FS_S(FS,X,T) = GDP_CGE_FS_S(FS,X,T); 

GDP_CGE_FS_S_Sum(X,T) = sum(FS_L, GDP_CGE_FS_S(FS_L,X,T)); 

 

* set 'com' shares equal to that of sim 

GDP_FS_S('com',X,T) = SIM_COM_S(X,T); 

GDP_FS_S_Sum(X,T) = 1-sum(FS_U, GDP_FS_S(FS_U,X,T)); 

 

* set 'linked' sector shares equal in proportion to what's in cge 

GDP_FS_S(FS_L,X,T) = GDP_CGE_FS_S(FS_L,X,T)/GDP_CGE_FS_S_Sum(X,T)*GDP_FS_S_Sum(X,T); 

 

* calculate sector GDP 

GDP_FS(FS,X,T) = GDP_FS_S(FS,X,T)*GDP(X,T); 

 

* Set Fuel Prices/Costs 

* Coal mining/exports (x/0.8+35 is washing costs and yield effect) 

* 0.95 on exports assumes a 5% royalty 

SIM_FUELP('MINCLE1',X,T) = SIM_CLE1(X,T); 

SIM_FUELP('MINCLE2',X,T) = SIM_CLE2(X,T); 

SIM_FUELP('MINCLE3',X,T) = SIM_CLE2(X,T)/0.8+35; 

SIM_FUELP('MINCLN',X,T)  = SIM_CLN(X,T); 

SIM_FUELP('MINCLE-A',X,T) = SIM_CLE_A(X,T); 

SIM_FUELP('MINCLN-A',X,T) = SIM_CLN_A(X,T); 

SIM_FUELP('MINCLS',X,T) = SIM_CLE1(X,T); 

SIM_FUELP('MINCLS-A',X,T) = SIM_CLN_A(X,T); 

SIM_FUELP('MINCOA',X,T) = SIM_FUELP('MINCLE3',X,T)/0.8+35; 

* Exports 

SIM_FUELP('PEXCME',X,T) = SIM_GCOAL(X,T)* 0.95 * XRATE(X,T) * (-1); 

 

* Convert coal prices from 2013 R/ton to 2010 R/GJ 

SIM_FUELP(FUELPC,X,T) = SIM_FUELP(FUELPC,X,T)/COAL_CV(FUELPC) / 1.172; 

 

* Oil and derivatives prices 

SIM_FUELP(FUELPO,X,T) = SIM_GOIL(X,T) * XRATE(X,T) * OCRFAC(FUELPO,'a') + OCRFAC(FUELPO,'b'); 

 

* Natural Gas Techs, +1, +0.5 for GWL and GRL are the transport costs 

* the 0.95 for the export price is a royalty fee 

SIM_FUELP('MINGIH',X,T) = SIM_GIH(X,T) * XRATE(X,T) / 1.055; 

SIM_FUELP('IMPGWL',X,T) = (SIM_GGAS(X,T)+1) * XRATE(X,T) / 1.055; 

SIM_FUELP('IMPGRL',X,T) = (SIM_GGAS(X,T)+0.5) * XRATE(X,T) / 1.055; 

SIM_FUELP('PEXGAS',X,T) = SIM_GGAS(X,T)*0.95 * XRATE(X,T) / 1.055 * (-1); 

 

* Allocate Power Plant Techs Sets - can't get this to work 

*ERSOLT(PRC)$(PWRSET(PRC) eq 5)=yes; 

*ERSOLP(PRC)$(PWRSET(PRC) eq 6)=yes; 

 



 

 

* Calculate Solar costs based on 2010 ratios and smpled data, assuming ratios stay the same 

* and converting to R/kW as needed by SATIM 

SIM_TECHC(ERSOLP,X,T) = (PV_CRATIO(ERSOLP) * SIM_PV_BOS(X,T) + SIM_PV_MODULE(X,T)) * 1000 * XRATE(X,T); 

SIM_TECHC(ERSOLT,X,T) = CSP_CRATIO(ERSOLT) * SIM_CSP(X,T) * 1000 * XRATE(X,T); 

 

* Calculate Nuclear Cost converting from 2010 $ to 2010 R 

SIM_TECHC(ETN,X,T) = SIM_NUCLEAR(X,T) * XRATE(X,T); 

 

 

 

 

scalars 

  EFVAL                 temporary values stored here 

  XLTEST 

; 

 

 

 

LOOP(X, 

 

* Processing of spreadsheet input before exporting to TIMES 

 

* Calculate Sectoral GDP 

 

  GDP_FSX(FS,T) = GDP_FS(FS,X,T); 

  POPX(T) = SIM_POP(X,T); 

 

  execute_unload "MCDem.gdx" POPX COM_PROJ_BY GDP_FSX 

$ontext 

  execute 'xlstalk.exe -m DMD_PRJ_MC.xlsm'; 

   XLTEST = ERRORLEVEL; 

  IF(XLTEST = 1, 

    execute 'xlstalk.exe -c DMD_PRJ_MC.xlsm'; 

  ); 

  IF(XLTEST = 2, 

    execute 'xlstalk.exe -s DMD_PRJ_MC.xlsm'; 

  ); 

$offtext 

  execute 'gdxxrw.exe i=MCDem.gdx o=DMD_PRJ_MC.xlsx index=index_G2E!a6'; 

 

*  execute 'xlstalk.exe -o DMD_PRJ_MC.xlsm'; 

 

  execute 'gdxxrw.exe i=DMD_PRJ_MC.xlsx o=mcdem2.gdx index=index_E2G!a6 checkdate' 

 

  execute_load "mcdem2.gdx" SIM_DEMX 

 

* Fuel Price DDS file 

 PUT  SIM_COMCOST_FILE; 

 SIM_COMCOST_FILE.pc = 2; 

 SIM_COMCOST_FILE.nd = 5; 

 SIM_COMCOST_FILE.ap = 0; 

 

 PUT 'PARAMETER AACT_COST /' /; 

 

LOOP((FUELP,T), 

  EFVAL = SIM_FUELP(FUELP,X,T); 

  if(EFVAL, 

    PUT "REGION1.", FUELP.TL, ".", T.TL, EFVAL /; 

  ); 

); 

 

PUTCLOSE "/;"; 

 

* Tech Cost DDS File 

 PUT  SIM_PRCCOST_FILE; 

 SIM_PRCCOST_FILE.pc = 2; 

 SIM_PRCCOST_FILE.nd = 5; 

 SIM_PRCCOST_FILE.ap = 0; 

 

PUT 'PARAMETER ANCAP_COST /' /; 

 

LOOP((TECHC,T), 

  EFVAL = SIM_TECHC(TECHC,X,T); 

  if(EFVAL, 

    PUT "REGION1.", TECHC.TL, ".", T.TL, EFVAL /; 

  ); 

); 

PUTCLOSE "/;"; 

 

*$ontext 

* Demand DDS File 

 PUT  SIM_DEM_FILE; 

 SIM_DEM_FILE.pc = 2; 

 SIM_DEM_FILE.nd = 5; 

 SIM_DEM_FILE.ap = 0; 

 

PUT 'PARAMETER ACOM_PROJ /' /; 

 

 

LOOP((DEM1,TC), 

  EFVAL = SIM_DEMX(DEM1,TC); 

  if(EFVAL, 

    PUT "REGION1.", DEM1.TL, ".", TC.TL, EFVAL /; 

ELSE 

    PUT "REGION1.", DEM1.TL, ".", TC.TL, "EPS" /; 

  ); 

); 

PUTCLOSE "/;"; 

*$offtext 

 



 

 

* Run TIMES model 

  execute ".\satm\gams_wrkti\RUNTIMES2.CMD" 

 

* Extract Results from TIMES run 

  execute_load ".\links\1timesoutput.gdx" TVCOMNET TVCAP TVACT TVACTELC TVCOMBAL TVCOMBALELC POBJz; 

*TVNCAP TVNCAPF TVCAP TVFLO TVACT NCAP_ILED OBJ_ICOST TVCOMBAL TVCST_INVC TVCST_ACTC TVCST_FIXC; 

 

* Emissions 

  SIMO_CO2S(CO2SET,X,T) = TVCOMNET(T,CO2SET); 

 

* Power plants 

* Capacity (GW) 

  SIMO_PWRCAP('ETC',X,T) = sum(ETC, TVCAP(T,ETC)); 

  SIMO_PWRCAP('ETCA',X,T) = sum(ETCA, TVCAP(T,ETCA)); 

  SIMO_PWRCAP('ETG',X,T) = sum(ETG, TVCAP(T,ETG)); 

  SIMO_PWRCAP('ETGIH',X,T) = sum(ETGIH, TVCAP(T,ETGIH)); 

  SIMO_PWRCAP('ETN',X,T) = sum(ETN, TVCAP(T,ETN)); 

  SIMO_PWRCAP('ERH',X,T) = sum(ERH, TVCAP(T,ERH)); 

  SIMO_PWRCAP('ERSOLPC',X,T) = sum(ERSOLPC, TVCAP(T,ERSOLPC)); 

  SIMO_PWRCAP('ERSOLPR',X,T) = sum(ERSOLPR, TVCAP(T,ERSOLPR)); 

  SIMO_PWRCAP('ERSOLT',X,T) = sum(ERSOLT, TVCAP(T,ERSOLT)); 

  SIMO_PWRCAP('ERW',X,T) = sum(ERW, TVCAP(T,ERW)); 

  SIMO_PWRCAP('ERB',X,T) = sum(ERB, TVCAP(T,ERB)); 

 

* Production (TWh) 

  SIMO_PWRACT('ETC',X,T) = sum(ETC, TVACT(T,ETC)); 

  SIMO_PWRACT('ETCA',X,T) = sum(ETCA, TVACT(T,ETCA)); 

  SIMO_PWRACT('ETG',X,T) = sum(ETG, TVACT(T,ETG)); 

  SIMO_PWRACT('ETGIH',X,T) = sum(ETGIH, TVACT(T,ETGIH)); 

  SIMO_PWRACT('ETN',X,T) = sum(ETN, TVACT(T,ETN)); 

  SIMO_PWRACT('ERH',X,T) = sum(ERH, TVACT(T,ERH)); 

  SIMO_PWRACT('ERSOLPC',X,T) = sum(ERSOLPC, TVACT(T,ERSOLPC)); 

  SIMO_PWRACT('ERSOLPR',X,T) = sum(ERSOLPR, TVACT(T,ERSOLPR)); 

  SIMO_PWRACT('ERSOLT',X,T) = sum(ERSOLT, TVACT(T,ERSOLT)); 

  SIMO_PWRACT('ERW',X,T) = sum(ERW, TVACT(T,ERW)); 

  SIMO_PWRACT('ERB',X,T) = sum(ERB, TVACT(T,ERB)); 

  SIMO_PWRACT(TECHS,X,T) = SIMO_PWRACT(TECHS,X,T)/3.6; 

 

* Coal Mine Production 

  SIMO_COAACT(FUELPC,X,T) = TVACT(T,FUELPC)/COAL_CV(FUELPC); 

* Coal Transfers from Waterberg to CB 

  SIMO_COATRA(X,T) = (TVACT(T,'UXLE-AB')+TVACT(T,'UXLE2-AB'))/COAL_CV('MINCLE1'); 

* Coal Price Calculation 

* Average over all plants (we need to subtract the coal used by other sectors) 

  SIMO_COAP(X,T) = (sum(FUELPCPWR, (TVACT(T,FUELPCPWR)*SIM_FUELP(FUELPCPWR,X,T)))+(SIMO_COATRA(X,T)*252/1.172)) 

                    / (sum(FUELPCPWR,TVACT(T,FUELPCPWR))/COAL_CV('MINCLE1')+SIMO_COATRA(X,T))*1.172; 

* Central Basin Price (we need to subtrac the caol used by other sectors) 

  SIMO_COAPB(X,T) = (sum(FUELPCPWR_CB, (TVACT(T,FUELPCPWR_CB)*SIM_FUELP(FUELPCPWR_CB,X,T)))+(SIMO_COATRA(X,T)*252/1.172)) 

                    / (sum(FUELPCPWR_CB,TVACT(T,FUELPCPWR_CB))/COAL_CV('MINCLE1')+SIMO_COATRA(X,T))*1.172; 

* Waterberg Price 

  SIMO_COAPA(X,T) = (sum(FUELPCPWR_A, (TVACT(T,FUELPCPWR_A)*SIM_FUELP(FUELPCPWR_A,X,T)))) 

                    / (sum(FUELPCPWR_A,TVACT(T,FUELPCPWR_A))/COAL_CV('MINCLE1'))*1.172; 

 

* Gas Production 

  SIMO_GASACT(FUELPG,X,T) = TVACT(T,FUELPG); 

 

* Final year demand 

  SIMO_DMD(DEM1,X) = SIM_DEMX(DEM1,'2050'); 

 

* Electricity demand TWh 

  SIMO_ELCDMD(T,X) = sum(S, TVACTELC(T,S))/3.6/0.961; 

* Electricity Peak demand GW 

  SIMO_PEAKELC(T,X) = TVCAP(T,'ETRANS')/0.961; 

* Electricity price 2013 R/kWh 

  SIMO_ELCP(T,X) = sum(S, TVACTELC(T,S)*TVCOMBALELC(T,S))*1.172*3.6/1000000; 

 

* Objective function 

  SIMO_OBJz(X) = POBJz; 

 

  Display X; 

 

); 

 


